

NAG Library Routine Document

C02AMF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

C02AMF determines the roots of a cubic equation with complex coefficients.

2 Specification

```
SUBROUTINE C02AMF (U, R, S, T, ZEROR, ZEROI, ERREST, IFAIL)
```

```
INTEGER IFAIL
REAL (KIND=nag_wp) ZEROR(3), ZEROI(3), ERREST(3)
COMPLEX (KIND=nag_wp) U, R, S, T
```

3 Description

C02AMF attempts to find the roots of the cubic equation

$$uz^3 + rz^2 + sz + t = 0,$$

where u , r , s and t are complex coefficients with $u \neq 0$. The roots are located by finding the eigenvalues of the associated 3 by 3 (upper Hessenberg) companion matrix H given by

$$H = \begin{pmatrix} 0 & 0 & -t/u \\ 1 & 0 & -s/u \\ 0 & 1 & -r/u \end{pmatrix}.$$

The eigenvalues are obtained by a call to F08PSF (ZHSEQR). Further details can be found in Section 8.

To obtain the roots of a quadratic equation, C02AHF can be used.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- | | | |
|----|--|--------------|
| 1: | U – COMPLEX (KIND=nag_wp)
<i>On entry:</i> u , the coefficient of z^3 .
<i>Constraint:</i> $U \neq (0.0, 0.0)$. | <i>Input</i> |
| 2: | R – COMPLEX (KIND=nag_wp)
<i>On entry:</i> r , the coefficient of z^2 . | <i>Input</i> |
| 3: | S – COMPLEX (KIND=nag_wp)
<i>On entry:</i> s , the coefficient of z . | <i>Input</i> |
| 4: | T – COMPLEX (KIND=nag_wp)
<i>On entry:</i> t , the constant coefficient. | <i>Input</i> |

- 5: ZEROR(3) – REAL (KIND=nag_wp) array Output
 6: ZEROI(3) – REAL (KIND=nag_wp) array Output

On exit: ZEROR(i) and ZEROI(i) contain the real and imaginary parts, respectively, of the i th root.

- 7: ERREST(3) – REAL (KIND=nag_wp) array Output

On exit: ERREST(i) contains an approximate error estimate for the i th root.

- 8: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $U = (0.0, 0.0)$.

IFAIL = 2

The companion matrix H cannot be formed without overflow.

IFAIL = 3

The iterative procedure used to determine the eigenvalues has failed to converge.

7 Accuracy

If IFAIL = 0 on exit, then the i th computed root should have approximately $|\log_{10}(\text{ERREST}(i))|$ correct significant digits.

8 Further Comments

The method used by the routine consists of the following steps, which are performed by routines from LAPACK in Chapter F08.

- (a) Form matrix H .
- (b) Apply a diagonal similarity transformation to H (to give H').
- (c) Calculate the eigenvalues and Schur factorization of H' .
- (d) Calculate the left and right eigenvectors of H' .
- (e) Estimate reciprocal condition numbers for all the eigenvalues of H' .
- (f) Calculate approximate error estimates for all the eigenvalues of H' (using the 1-norm).

9 Example

This example finds the roots of the cubic equation

$$z^3 - (2 - 3i)z^2 + (5 + 14i)z - (40 + 5i) = 0.$$

9.1 Program Text

```

Program c02amfe

!      CO2AMF Example Program Text
!
!      Mark 24 Release. NAG Copyright 2012.
!
!      .. Use Statements ..
Use nag_library, Only: c02amf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Complex (Kind=nag_wp)       :: r, s, t, u
Integer                      :: i, ifail
!      .. Local Arrays ..
Real (Kind=nag_wp)          :: errest(3), zeroi(3), zeror(3)
!      .. Executable Statements ..
Write (nout,*) 'CO2AMF Example Program Results'

!      Skip heading in data file
Read (nin,*)

Read (nin,*) u, r, s, t

ifail = 0
Call c02amf(u,r,s,t,zeror,zeroi,errest,ifail)

Write (nout,*)
Write (nout,*) ' Roots of cubic equation      ', &
'          Error estimates'
Write (nout,*) '          ', &
'          (machine-dependent)'
Write (nout,*)

Do i = 1, 3
  Write (nout,99999) ' z = ', zeror(i), zeroi(i), '*i', errest(i)
End Do

99999 Format (1X,A,1P,E12.4,Sp,E12.4,A,8X,SS,E9.1)
End Program c02amfe

```

9.2 Program Data

```

CO2AMF Example Program Data
( 1.0, 0.0)
(-2.0, 3.0)
( 5.0, 14.0)
(-40.0, -5.0) : Values of U, R, S and T

```

9.3 Program Results

CO2AMF Example Program Results

Roots of cubic equation	Error estimates (machine-dependent)
z = -2.0000E+00 +3.0000E+00*i	1.7E-15
z = 1.0000E+00 -2.0000E+00*i	3.6E-15
z = 3.0000E+00 -4.0000E+00*i	3.7E-15